



VIC

da zero

Sesta parte

di Tommaso Pantuso

Al telefono con VICTEL

Questo mese vi presentiamo un'applicazione pratica basata su gran parte delle nozioni apprese finora sull'uso del VIA, adattatore versatile d'interfaccia, contenuto all'interno del VIC 20 e che funge, per chi ancora non lo sapesse, da elemento di comunicazione con il mondo esterno tramite le sue due porte d'ingresso/uscita ad otto bit, da ricetrasmittente di dati seriali per mezzo dello shift register contenuto al suo interno e da unità di temporizzazione grazie a due timer i quali, ricordiamo, permettono sia la generazione di treni d'impulsi e di impulsi singoli di durata programmabile, che la misura del numero d'impulsi contenuti in un treno e la durata di ciascuno di essi. Da quanto riassunto si capisce bene il perché dell'attributo "versatile" dato al 6522 (che è appunto il VIA di cui stiamo parlando).

L'essere un'applicazione dimostrativa non toglie al circuito che andiamo a presentare la propria validità ed utilità nella vita quotidiana, come potrete voi stessi constatare se deciderete di montare e di utilizzare il VICTEL. Inoltre la semplicità del circuito ne permette l'uso, con la modifica del solo connettore d'ingresso, supportato da qualunque macchina che possieda almeno una linea di comunicazione con l'esterno, eventualità non molto difficile da riscontrare in pratica.

Cos'è il VICTEL?

È un circuito capace di generare delle sequenze codificate rappresentanti i numeri telefonici che, inviate alla centrale tramite la linea del telefono, ci permettono di contattare "l'utente desiderato" come direbbe la signorina della SIP.

La parte hardware è estremamente semplice in quanto composta da un transistor NPN che funge da driver per un relé da 60-100 ohm. Ricordiamo infatti che non è molto salutare per il VIA collegare su una sua linea direttamente il relé da comandare in quanto si rischierebbe di metterlo ben presto fuori uso, richiedendo il relé corrente superiore a quella erogabile dall'integrato. Si usa allora l'accorgimento di interporre tra il VIA ed il relé un transistor che separa i due elementi: la corrente viene in

questo caso fornita dal transistor che la preleva da quella di alimentazione.

Prima di dare ulteriori chiarimenti circa la realizzazione pratica pensiamo sia necessario spiegare, a grandi linee, cosa succede quando alziamo il ricevitore e ruotiamo il disco combinatore installato sul nostro apparecchio per comporre il numero.

Il disco combinatore serve per interrompere periodicamente la continuità della corrente che percorre la linea telefonica e ciò equivale a lanciare sulla stessa degli impulsi corrispondenti al numero che si desidera comporre (figura 1).

Quando solleviamo il microtelefono (la famosa cornetta) sulla linea passa una corrente ed in centrale viene azionato un pre-selettore che effettua la ricerca di un primo selettore di gruppo libero e si ferma quando quest'ultimo è stato agganciato (cfr. figura 2). Composta la prima cifra il primo selettore di gruppo (SG1) ruota alla ricerca di un secondo selettore di gruppo libero (SG2) nel gruppo corrispondente alla cifra che è stata selezionata e così via per ciascuna cifra fino al selettore finale di linea (SFL) che seleziona l'utente desiderato in corrispondenza dell'ultima cifra ricevuta.

Se il numero *chiamato* risulta libero viene inviata sulla linea una corrente alternata che sollecita la bobina dell'elettromagnete del campanello posto all'interno dell'apparecchio, provocando lo squillo; in caso contrario viene inviato un segnale di occupato che è percepito da chi ha effettuato la chiamata.

Anche se non molto rigorosi speriamo almeno di essere stati chiari sui punti precedentemente esposti. Si capisce quindi che tutte le operazioni in centrale vengono comandate dal nostro combinatore a disco e più precisamente dagli impulsi che esso invia sulla linea. Tali impulsi non sono di




Figura 1 - Se nel circuito schematicizzato in 1a) pigiamo e rilasciamo continuamente il pulsante P, passerà una corrente nel circuito quando P è premuto e non passerà corrente quando P viene lasciato libero. Se andiamo a misurare la tensione tra i punti H e K, rileveremo un segnale che ha l'andamento teorico rappresentato nella sezione 1b). Tramite il disco combinatore del telefono si ottiene lo stesso effetto.




Figura 2 - Alzando il microtelefono a casa, in centrale l'organo PR ruota fino ad agganciarsi con il selettore SG1. Possiamo a questo punto comporre le varie cifre ciascuna delle quali abiliterà un selettore. Per esempio la cifra "4", corrispondente a quattro impulsi inviati sulla linea, farà ruotare in un certo modo il primo selettore di gruppo SG1. Se facciamo seguire a questa un'altra cifra, inviando degli altri impulsi, questi passeranno attraverso SG1 e, contati da SG2, disporranno quest'ultimo in una determinata posizione. Tale processo si ripete per ogni cifra fino a che l'utente non è agganciato tramite il selettore finale di linea SFL.




Figura 3 - Tempi di un segnale codificante una parte di una sequenza telefonica. I contatti del disco combinatorio aprono la linea per 40 mS e la tengono chiusa per 60 mS per ogni unità costituente la cifra (ad esempio per il "5" questo processo viene ripetuto cinque volte).

durata casuale ma hanno dei cicli ben stabiliti.

Cominciamo col dire che al numero "1" corrisponde l'invio di un solo impulso; al numero "2" l'invio di due impulsi e così via fino al "9". Allo "0" non corrispondono, come erroneamente si potrebbe pensare, zero impulsi ma dieci.

Tali impulsi devono avere (SIP) un rapporto *mark-space* 40/60. In altre parole, essi devono avere la forma rappresentata in figura 3.

Se per esempio allora volessimo inviare sulla linea il numero 3, dovremmo interrompere la linea per 40 millisecondi e ricollegarla per 60 eseguendo tale operazione per tre volte (cioè è fatto automaticamente dal disco combinatorio). Occorre però una ulteriore informazione affinché gli organi della centrale possano discriminare una cifra dall'altra. Tale informazione è fornita da una pausa più lunga posta tra una cifra e l'altra detta *pausa interdigitale*. Per la SIP tale pausa, durante la quale la linea non viene interdetta, deve avere la durata di 500 ± 50 millisecondi. Secondo la SIP significa che rispettando tali standard è sicuro il corretto funzionamento degli organi della centrale.

Si possono verificare però casi in cui il funzionamento avvenga correttamente, cioè il numero chiamato venga agganciato, anche introducendo pause di separazione più brevi (cioè equivale a comporre il numero completo più rapidamente). Il prototipo da noi collaudato in alcune zone agganciava correttamente con una pausa di soli 200 millisecondi.

Nelle figure 4, 5 e 6 si possono osservare lo schema elettrico ed i disegni dello stampato




Figura 4 - Schema elettrico del circuito di comando. La linea si chiude da una parte sull'apparecchio telefonico e dall'altra va verso la centrale. La tensione di lavoro del relé da noi usato è di 6 volt.




Foto della scheda VICTEL.




Figura 5 - Vista delle piste del lato inferiore del circuito stampato che alloggia i componenti del VICTEL in grandezza naturale.





Figura 6 - Lato superiore del circuito stampato. Su tale circuito si può saldare un connettore per connetterlo alla user port oppure può essere inserito (dall'altra parte) nel bus VL1 del VICLAB. I triangolini indicano i pin corrispondenti ai capi della bobina del relé.

pato relativi al VICTEL mentre in figura 7 (pag. 106) è riportato il piano di cablaggio dei componenti. Per evitare confusione specifichiamo che R1 è la resistenza da 2200 ohm mentre R2 è quella da 4700 ohm. La conformazione dello stampato permette due tipi di collegamento: infatti chi possiede il bus VL1-VICLAB (MC n° 26) può inserire il VICTEL nell'apposito connettore posto su tale bus. Chi invece non possiede il VICLAB può inserire direttamente la scheda nella user port del VIC 20 tramite il connettore CN1. Si comprende quindi che tale connettore non è necessario per i possessori della VL1 che quindi possono eliminarlo.

Come già detto l'hardware è molto semplice e non richiede ulteriori commenti. La linea telefonica viene posta in serie al contatto del relé e quindi essa è interrotta o no a seconda della posizione di tale contatto.

Il relé è del tipo *dual in line package*, cioè i suoi piedini sono geometricamente disposti come quelli di un comune circuito integrato 8 + 8 (quindi potrebbe essere montato benissimo su uno zoccolo porta integrati) ed è a 6 volt.



Aggiungiamo che durante la composizione del numero si avvertirà nel ricevitore un po' di rumore dovuto alla scansione delle cifre.

Questo si sarebbe potuto evitare introducendo un ulteriore relé che avrebbe dovuto scollegare il ricevitore durante la composizione del numero ma non essendo il rumore eccessivamente fastidioso lasciamo questa modifica a chi è di orecchie molto delicate.

Chi volesse realizzare un apparecchio telefonico personalizzato utilizzando il VICTEL unitamente al microtelefono di un apparecchio citofonico, farà bene a collegare sul ricevitore due diodi ed una resistenza come indicato in figura 8 al fine di sopprimere notevolmente la rumorosità dovuta alla cifra in composizione.

Le funzioni

Il programma di comando che vi proponiamo non si limita solamente a comporre il numero telefonico desiderato ma effettua




Figura 8 - Chi vuole utilizzare il VICTEL collegandolo ad un apparecchio citofonico deve porre in parallelo al ricevitore r due diodi al silicio ed una resistenza 680-1000 ohm per ridurre il rumore ottenuto durante la composizione della cifra come indicato in questa figura.

altre sette funzioni e precisamente le seguenti:

- 1) M - Ripetizione
- 2) C - Cancellazione
- 3) T - Start timer
- 4) S - Stop timer
- 5) L - Reset linea
- 6) R - Reset programma
- 7) B - Buzzer

Descriviamo le suddette funzioni singolarmente secondo le operazioni che svolgono. Il modo in cui esse adempiono ai loro compiti sarà spiegato nella descrizione del programma.

M - Ripetizione: serve per ripetere un numero selezionato e risultato occupato. Il numero composto viene infatti automaticamente memorizzato in macchina.

C - Cancellazione: serve per cancellare un numero che si trova in memoria. Tale operazione dovrebbe essere ripetuta ogni

qual volta si vuol comporre un nuovo numero.

T - Start timer: visualizza un orologio tramite il quale è possibile misurare la durata della conversazione.

S - Stop timer: arresta l'orologio.

L - Reset linea: equivale a premere il tasto d'interruzione telefonica; viene usato quando si è trovato il numero chiamato occupato e si vuole richiamare oppure quando si vuole interrompere una comunicazione.

R - Reset programma: riporta sullo schermo il menu delle funzioni, ripristina la linea e azzerla la memoria telefonica.

B - Buzzer: alzando il microtelefono e premendo il tasto B, se un altro apparecchio telefonico è collegato sulla stessa linea in parallelo esso produrrà uno squillo; inoltre verrà escluso il segnale di linea permettendo così di utilizzare i due apparecchi come sistema interfonico.

Forse è inutile aggiungere che tutte queste operazioni potranno essere effettuate solo se il microtelefono è sollevato.

Il software

Vediamo come sia possibile generare delle cifre telefoniche tramite l'uso del computer e del VIA.

Gli impulsi con le giuste temporizzazioni vengono generati con una tecnica mista, cioè usando sia un ritardo hardware fornito da un timer interno del 6522 che un ritardo software generato dal programma. Lo schema di principio per la generazione di una cifra è indicato nel diagramma di figura 9.

Per prima cosa viene predisposto il registro di controllo ausiliario ACR in modo che gli impulsi generati per mezzo del timer T1 vengano inviati in uscita a comandare la base del transistor collegata tramite una resistenza alla linea PB7. Poi vengono




Figura 9 - Diagramma per la generazione della cifra corrispondente al valore numerico di AS.

caricati i latch basso ed alto di T1 con un valore numerico corrispondente ad un ritardo di 40 millisecondi.

Ricordiamo che T1 è mappato in memoria alle locazioni decimali indicate di seguito:

37140	contatore	:byte basso
37141		:byte alto
37142	latch	:byte basso
37143		:byte alto

Il funzionamento è il seguente: vengono prima caricati il latch basso e quello alto del timer con il valore del ritardo che si vuole ottenere; quando si vuol avviare il contatore basterà caricare nel byte alto del contatore lo stesso valore che si trova memorizzato nel latch alto. Il numero contenuto nel contatore sarà decrementato di uno ad ogni ciclo macchina: quando esso sarà zero, verrà alzato il flag relativo a T1 nel registro IFR del VIA.

Facciamo un esempio per chiarire le idee.

Se carichiamo il timer con il numero 40000, esso sarà decrementato di un'unità ad ogni ciclo macchina per cui il contatore sarà azzerato dopo 40000 cicli (in realtà dopo 40002). Se ora supponiamo che ogni ciclo sia di 1 microsecondo, cioè 0.000001 secondi, alla fine del conteggio avremo ottenuto un ritardo di $40000 \times 0.000001 =$

0.04 secondi, cioè 40 millisecondi.

Si tenga presente che per il VIC ogni ciclo è maggiore di 1.1 microsecondi.

Il valore 40000 decimale, in esadecimale è 9C40; il byte alto di tale numero è 9C e quello basso è 40. Per inserire tali valori in macchina con l'istruzione POKE dovremo, come già sapete, codificare ciascun byte in decimale (9C = 156 decimale e 40 = 64 decimale) quindi introdurre i valori ottenuti nei byte alto e basso del latch del timer con le istruzioni POKE 37143,156 e POKE 37142,64.

Ritornando al nostro diagramma, una volta caricato T1, il programma legge il valore della cifra da noi introdotta e trasformata per comodità in una stringa, dopo di che carica il contatore alto del timer generando un primo ritardo di 40 millisecondi; viene quindi generato un ritardo software di 60 millisecondi tramite un ciclo di FOR... NEXT ripetendo il tutto per un numero di volte corrispondente al valore della cifra introdotta generando, alla fine di ogni ciclo completo, la pausa di separazione del valore di 500 millisecondi. Il tutto viene ripetuto per ciascuna cifra.

Nel listato 1 si può leggere la trasposizione in programma dell'algoritmo di figura 9. Chi non ha bisogno quindi di un programma che svolga tutte le funzioni del VICTEL può prendere spunto da quest'ul-

```

10 PRINT "J":PRINT"COMPORRE IL NUMERO"
20 GETA$:IFA$=""THEN20
30 IFA$="0"THENA$="10"
40 POKE37147,128      :REM ACR
50 POKE37142,0        :REM LATCH BASSO
60 POKE37143,175     :REM LATCH ALTO
70 FORS=1TOVAL(A$)
80 POKE37141,175     :REM AVVIO CONTEGGIO
90 FORI=1TO70:NEXTI  :REM RITARDO 60 MS
100 NEXTS
110 IFA$="10"THENA$="0"
120 PRINTA$;
130 FORN=1TO500:NEXTN:REM PAUSA 500 MS
140 GOTO20

```

Listato 1 - Nella linea 90 non omettete l a chiusura del ciclo di FOR... NEXT poiché il ciclo diventerebbe più veloce e cambierebbero le temporizzazioni previste.

timo ed ampliarlo secondo le proprie esigenze o lasciarlo addirittura invariato. Il valore 175-0 posto nel contatore permette di generare la forma d'onda evidenziata in figura 10 (pag. 108). Il valore 500 MS indicato alla riga 130 è approssimativo.

Veniamo dunque alla descrizione del programma del listato 2 che abbiamo denominato VICTEL.

Esso è stato dimensionato per memorizzare un numero massimo di 15 cifre. Questo numero è però facilmente estensibile modificando semplicemente la dimensione del vettore AS(I) nella riga 50. Esaminiamo le varie subroutine in dettaglio.

Dalla riga 270 alla 290 abbiamo il posi-

Listato 2

```

10 REM +-----+
20 REM VICTEL - COPYRIGHT 1984 - TOMMASO PANTUSO
30 REM +-----+
40 REM
50 PRINT "J":DIM A$(15):N=0:POKE36879,25
60 PRINT "VICTEL - (C) TP 1984" PRINT
70 PRINT "M# COMPORRE IL NUMERO # TELEFONICO DESIDERATO#":PRINT,PRINT
80 PRINT "M# RIPETIZIONE #":PRINT
90 PRINT "C# CANCELLAZIONE #":PRINT
100 PRINT "T# START TIMER #":PRINT
110 PRINT "S# STOP TIMER #":PRINT
120 PRINT "L# RESET LINEA #":PRINT
130 PRINT "R# RESET PROGRAMMA #":PRINT
140 PRINT "B# BUZZER #":PRINT
150 GETA$:IFA$=""THEN150
160 IFA$="M"THENPRINTCHR$(13):POKE37147,128:GOSUB390:GOTO150
170 IFA$="C"THENCLR:POKE37147,64:PRINTCHR$(67):PRINTCHR$(17):GOTO150
180 IFA$="T"THEN620
190 IFA$="L"THENPOKE37147,64:GOSUB0:GOTO150
200 IFA$="R"THENPRINT "J":CLR:POKE37147,64:GOSUB0:GOTO70
210 IFA$="B"THENGOSUB780:GOTO150
220 IFA$="R"THENASC(A$)>57THEN150
230 IFA$="0"THENA$="10"
240 REM -----
250 REM POSIZIONAMENTO DI ACR E CARICAMENTO DEI LATCHS H ED L
260 REM -----
270 POKE37147,128
280 POKE37142,0
290 POKE37143,175:GOSUB470
300 REM -----
310 REM TRASFORMA LA VARIABILE A$ IN UN ELEMENTO DI UN VETTORE
320 REM -----
330 N=N+1:A$(N)=A$:GOSUB560
340 GOSUB540
350 GOTO150
360 REM -----
370 REM CICLO RIPETIZIONE NUMERO DOPO LA PRESSIONE DEL TASTO M
380 REM -----
390 FORL=1TON
400 A$=A$(L)
410 GOSUB470:GOSUB540:GOSUB560
420 NEXTL
430 RETURN
440 REM -----
450 REM GENERA UN NUMERO D'IMPULSI 40/60 PARI A VAL(A$)
460 REM -----
470 FORS=1TOVAL(A$)
480 POKE37141,175:FORI=1TO70:NEXTI
490 NEXTS
500 RETURN
510 REM -----
520 REM GENERA LA PAUSA INTERDIGITALE E VISUALIZZA IL NUMERO COMPOSTO
530 REM -----
540 FORI=1TO500:NEXTI
550 RETURN
560 IFA$="10"THENA$="0"
570 PRINTA$;
580 RETURN
590 REM -----
600 REM CRONOMETRO
610 REM -----
620 TI$="000000"
630 POKE36879,25:P$="" PRINT
640 PRINT "-----"
650 PRINT "H M S"
660 PRINT "-----"
670 H=MID$(TI$,1,2)+P$+MID$(TI$,3,2)+P$+MID$(TI$,5,2)
680 IFPEEK(203)=10THENPRINT "J":GOTO70
690 PRINTH$
700 PRINT "-----TT"
710 H=VAL(MID$(H$,7,2))
720 IFINT(H/30)-H/30=0THEN GOSUB 960
730 IFPEEK(203)=41THENPRINT "X":GOTO150
740 GOTO670
750 REM -----
760 REM BUZZER
770 REM -----
780 POKE37147,128
790 POKE37142,0
800 POKE37143,180
810 FORS=1TO10
820 POKE37141,180:FORI=1TO 50:NEXTI,S:POKE37147,64
830 RETURN
840 REM -----
850 REM RIPRISTINO LINEA
860 REM -----
870 POKE37136,128
880 POKE37136,0
890 FORI=1TO1000:NEXT
900 POKE37136,128
910 POKE37136,0
920 RETURN
930 REM -----
940 REM BEEP
950 REM -----
960 POKE36878,15:POKE36876,230
970 FORI=1TO200:NEXT:POKE36878,0
980 RETURN
990 REM - "J" = SHIFT+CLR HOME
1000 REM - "R" = CTRL+6 & CTRL+9
1010 REM - "M" = CTRL+7 & CTRL+8
1020 REM - "T" = SHIFT+CRSR
1030 REM - "X" = CRSR

```




Figura 10 - Oscillogramma rilevato su PB7. Ogni divisione orizzontale equivale a 20 mS.

zionamento di ACR ed il caricamento dei latch del timer TI.

Ricordiamo che il bit 7 di ACR abilita o disabilita l'uscita su PB7 comandata da TI a seconda che si trovi in condizione 1 o 0. Il bit 6 abilita invece tale uscita a funzionare in modo da generare singoli impulsi o treni d'impulsi a seconda che esso si trovi allo stato 0 od 1.

Da 330 a 350 avviene la trasformazione della cifra, che di volta in volta è introdotta, nell'elemento costituente di un vettore che viene memorizzato per poter poi effettuare la ripetizione del numero automaticamente. Questa è appunto la funzione che viene svolta dalla riga 490 alla 420. Infatti questo segmento ripescia in memoria i vari elementi del vettore componente il numero e li invia alla routine di composizione del numero stesso (470, 540, 560).

Da 470 a 490 viene caricato il byte alto del contatore di TI e generato lo spazio per un numero di volte pari al valore della cifra introdotta.

Da 540 a 580 viene generata la pausa che

divide una cifra dall'altra ed ogni cifra viene impressa sullo schermo. Se volete aumentare la velocità di composizione del numero provate a cambiare il valore contenuto nel ciclo FOR... NEXT della linea 540. Noi abbiamo rilevato un funzionamento corretto fino ad un valore 200; in ogni caso 500 è il valore che si avvicina di più allo standard. Da riga 620 a riga 740 è generato sullo schermo l'orologio per il controllo della durata della conversazione sfruttando la variabile TIS fornita dalla macchina. Poiché le sei cifre da cui essa è formata sono visualizzate tutte adiacenti, vengono usate alcune istruzioni di stringa per suddividerle in gruppi di due separate da uno spazio. Le linee 720 e 730 di tale segmento effettuano un controllo sull'incremento di TIS provocando un beep ogni 30 secondi (GOSUB 960). Anche questo valore può essere facilmente variato. Vi facciamo notare che durante il beep si può constatare una discontinuità di impressione sullo schermo. Non preoccupatevi di ciò in quanto tale fenomeno si verifica poiché quando è in funzione il generatore di

nota, non può essere visualizzato nessun dato. Questo non vieta comunque alla macchina di aggiornare continuamente TIS in quanto ciò fa parte dei compiti delle routine di manipolazione degli interrupt. Un'altra cosa che osserverete sicuramente è che nonostante sia stato programmato un solo beep, ne udirete più di uno. Questo avviene perché la variabile TIS permane nella condizione favorevole al test della linea 720 per un tempo superiore a quello occorrente ad effettuare il test stesso, ad andare alla subroutine beep ed a ritornare. In altre parole quando il programma torna dalla subroutine beep, ritrova TIS nelle condizioni in cui è stata lasciata in quanto tale variabile viene aggiornata ogni secondo mentre il programma impiega molto meno tempo per le verifiche.

Dalla riga 780 alla 820 viene invece inviato sulla linea un numero simile ad uno zero per far squillare un apparecchio telefonico posto in parallelo sulla linea dopo la pressione del tasto "B". La routine di ripristino linea manda su PB7, tramite l'uso delle porte d'I/O prima uno "zero" e poi un "uno" corrispondenti ad una tensione di 0 volt e di 5 volt tra cui è interposta una pausa. Tale operazione diseccita il relè e lo rieccita dopo un certo tempo provocando una momentanea interruzione della linea. Questa sezione viene usata anche durante l'operazione di reset programma con l'aggiunta della pulizia del video, della rivisualizzazione del menu e dell'azzeramento del numero in memoria.

Non rimane più molto da dire sul software se non che dopo il RUN apparirà sullo schermo il menu delle varie operazioni che saranno avviate dopo la pressione del tasto corrispondente. Ricordate che prima di comporre un nuovo numero è meglio cancellare quello precedente per evitare che il nuovo numero si aggiunga a quello già contenuto in memoria.

Questo programma può essere ampliato aggiungendo una parte che serva da rubrica telefonica e da archivio il quale ricerchi un numero in esso depositato a partire da un nome o da un codice risolvendo non pochi problemi a chi debba usare frequentemente il telefono. Se comunque avete dei suggerimenti da darci fateceli pervenire.


E questo è tutto per ciò che riguarda il VICTEL.




E chi non ha un VIC?

Data l'estrema semplicità del circuito presentato è possibile adattarlo ad una qualunque macchina dalla quale si possa prelevare un segnale di controllo. Si può usare a questo proposito il circuito stampato presentato in questo riquadro collegando le terminazioni del computer ai punti contrassegnati con "+", "m" e "p".

Sul primo terminale va collegata l'alimentazione a +5 volt; il secondo va collegato a massa; sul terzo si manda il segnale per il comando del relè. Se non si è in grado di generare il segnale necessario per il controllo tramite un timer perché non presente, esso può essere molto facilmente ottenuto facendo variare il segnale di comando alternativamente da alto a basso secondo le temporizzazioni ben definite dallo standard. Presentiamo anche un diagramma per la codificazione delle cifre mediante tale tecnica. In tale diagramma N rappresenta la singola cifra del numero.



Circuito Stampato visto dalla parte del rame.



Casio PB 700

un completo sistema personal nello spazio di un pocket

Euro-Advertising



CASIO PB 700: in un unico complesso estremamente compatto, un grande display grafico a 4 linee, una memoria veramente capace, una plotter-stampante a 4 colori ed un micro-registratore a cassetta; tutto ciò per rendere il PB 700 un vero computer da tavolo facilmente trasportabile in una valigetta.

La grande e sperimentata tecnologia CASIO significa anche Computer, Calcolatrici, Registratori di Cassa.



FP 1000 (16 K Bytes)

FP 200

Dotato di un chiaro manuale per la programmazione in Basic è destinato a diventare collaboratore inseparabile di chi studia o lavora sia in campo tecnico che commerciale. Compatibile con unità più grandi della gamma Casio.

- RAM 4k bytes espandibile a 16k bytes;
- Display grafico a 4 linee x 20 colonne;
- Dimensioni: mm. 23x200x88.

In vendita nei negozi specializzati tra i quali: CLUP, VIGANÒ, SALMOIRAGHI, BIT SHOP PRIMAVERA.

CASIO

DITRON
SpA

Per informazioni più dettagliate scrivere a Ditron S.p.A. Via-
le Certosa, 138 - Milano.

NOME _____

COGNOME _____

VIA _____

CITTA _____

C.A.P. _____

MC _____